

SUPPLEMENTARY MATERIAL 3. JAGS syntax for top-ranked negative binomial regression model relating habitat and Greater Sage-Grouse-related variables to the abundance of western harvester ant mounds in central Wyoming, USA. The model is specified as a hierarchical Poisson-Gamma mixture model in a Bayesian framework. All predictor variables were standardized to have a mean of 0 and standard deviation of 1 prior to model fitting.

```

model{
# Likelihood
for(i in 1:nunits){
  count[i]~dpois(lambda[i])
  lambda[i]<-phi[i]*rho[i]
  log(phi[i])<-beta[1] + beta[2]*herb[i] + beta[3]*shrub[i] + beta[4]*depth[i] + beta[5]*sand[i] +
beta[6]*grouseindex[i] + beta[7]*log2pellets[i]
  rho[i]~dgamma(theta,theta)
}

# Priors
theta~dunif(0,10)

beta[1:7]~dmnorm(muB[,],varB[,])
muB[1]<- 0.5 # intercept
muB[2]<- 1.5 # herb
muB[3]<- -1.5 # shrub
muB[4]<- 1.5 # depth
muB[5]<- 1.5 # sand
muB[6]<- -1.5 # grouseindex
muB[7]<- 1.5 # log2pellets

varB[1,1]<-0.1; varB[1,2]<-0; varB[1,3]<-0; varB[1,4]<-0; varB[1,5]<-0; varB[1,6]<-0; varB[1,7]<-0
varB[2,1]<-0; varB[2,2]<-0.1; varB[2,3]<-0; varB[2,4]<-0; varB[2,5]<-0; varB[2,6]<-0; varB[2,7]<-0
varB[3,1]<-0; varB[3,2]<-0; varB[3,3]<-0.1; varB[3,4]<-0; varB[3,5]<-0; varB[3,6]<-0; varB[3,7]<-0
varB[4,1]<-0; varB[4,2]<-0; varB[4,3]<-0; varB[4,4]<-0.1; varB[4,5]<-0; varB[4,6]<-0; varB[4,7]<-0
varB[5,1]<-0; varB[5,2]<-0; varB[5,3]<-0; varB[5,4]<-0; varB[5,5]<-0.1; varB[5,6]<-0; varB[5,7]<-0
varB[6,1]<-0; varB[6,2]<-0; varB[6,3]<-0; varB[6,4]<-0; varB[6,5]<-0; varB[6,6]<-0.1; varB[6,7]<-0
varB[7,1]<-0; varB[7,2]<-0; varB[7,3]<-0; varB[7,4]<-0; varB[7,5]<-0; varB[7,6]<-0; varB[7,7]<-0.1

# Bayesian p-value (Dp) for goodness-of-fit
for(i in 1:nunits){
  ystar[i]~dpois(lambda[i])
  D[i]<-(count[i]-lambda[i])*(count[i]-lambda[i])
  Dstar[i]<-(ystar[i]-lambda[i])*(ystar[i]-lambda[i])
  p_val[i]<-step(Dstar[i]-D[i])
}
sumDstar<-sum(Dstar[])
sumD<-sum(D[])
Dp<-step(sumDstar-sumD)
}

```